EXAMPLE 3.1-2: Sensor in an Oscillating Temperature Environment

A temperature sensor is installed in a chemical reactor that operates in a cyclic fashion. The temperature of the fluid in the reactor varies in an approximately sinusoidal manner with a mean temperature $\overline{T}_{\infty}$ =320°C, an amplitude ΔT_{∞} =50°C, and a frequency f = 0.5 Hz. The sensor can be modeled as a sphere with diameter D = 1.0 mm. The sensor is made of a material with conductivity k_s = 50 W/m-K, specific heat capacity c_s = 150 J/kg-K, and density ρ_s = 16000 kg/m^3. In order to provide corrosion resistance, the sensor has been coated with a thin layer of plastic; the coating is th_c=100 μm thick with conductivity k_c=0.2 W/m-K and has negligible heat capacity relative to the sensor itself. The heat transfer coefficient between the surface of the sensor and the fluid is $\overline{h}$ =500 W/m^2-K. The sensor is initially at T_{ini} = 260°C.

a.) Is a lumped capacitance model of the temperature sensor appropriate?

The inputs are entered in EES:

```
"EXAMPLE 3.1-2: Sensor in an Oscillating Temperature Environment"
$UnitSystem SI MASS RAD PA  K J
$TABSTOPS   0.2 0.4 0.6 0.8 3.5 in

"Inputs"
T_infinity_bar=converttemp(C,K,320[C])          "average temperature of reactor"
T_ini=converttemp(C,K,260[C])                   "initial temperature of sensor"
DT_infinity=50[K]                               "amplitude of reactor temperature change"
f=0.5 [Hz]                                      "frequency of reactor temperature change"
D=1.0 [mm]*convert(mm,m)                        "diameter of sensor"
k_s=50 [W/m-K]                                  "conductivity of sensor material"
c_s=150 [J/kg-K]                                "specific heat capacity of sensor material"
rho_s=16000 [kg/m^3]                            "density of sensor material"
th_c=100 [micron]*convert(micron,m)             "thickness of coating"
k_c=0.2 [W/m-K]                                 "conductivity of coating"
h_bar=500 [W/m^2-K]                             "heat transfer coefficient"
```

The Biot number is the ratio of the internal conduction resistance to the resistance to heat transfer from the surface of the object. In this problem, the resistance to heat transfer from the surface is the series combination of convection (R_{conv}):

$$R_{conv} = \frac{1}{\overline{h}\, 4\,\pi \left(\frac{D}{2} + th_c\right)^2}$$

and the conduction resistance of the coating ($R_{cond,c}$, from Table 1-2):

$$R_{cond,c} = \frac{\left[\frac{2}{D} - \frac{2}{\left(D + 2\,th_c\right)}\right]}{4\,\pi\, k_c}$$

```
R_conv=1/(h_bar*4*pi*(D/2+th_c)^2)                    "convective resistance"
R_cond_c=(1/(D/2)-1/(D/2+th_c))/(4*pi*k_c)            "conduction resistance of coating"
```

The resistance to internal conduction ($R_{cond,int}$) is approximated according to:

$$R_{cond,int} = \frac{L_{cond}}{k_s\, A_s}$$

where A_s is the surface area of the sensor:

$$A_s = 4\pi\left(\frac{D}{2}\right)^2$$

and L_{cond} is the conduction length, approximated according to Eq. (3-2):

$$L_{cond} = \frac{V}{A_s}$$

where:

$$V = \frac{4\pi}{3}\left(\frac{D}{2}\right)^3$$

```
V=4*pi*(D/2)^3/3                          "volume of sensor"
A_s=4*pi*(D/2)^2                          "surface area of sensor"
L_cond=V/A_s                              "approximate conduction length"
R_cond_int=L_cond/(k_s*A_s)               "internal conduction resistance"
```

The Biot number that characterizes this problem is therefore:

$$Bi = \frac{R_{cond}}{R_c + R_{conv}}$$

```
Bi=R_cond_int/(R_conv+R_cond_c)       "Biot number"
```

which leads to Bi = 0.0018; this is sufficiently less than 1 to justify a lumped capacitance model.

b.) What is the time constant associated with the sensor? Do you expect there to be a substantial temperature measurement error related to the dynamic response of the sensor?

The time constant (τ_{lumped}) is the product of the resistance to heat transfer from the surface of the sensor (which is related to conduction through the coating and convection) and the thermal mass of the sensor (C):

$$\tau_{lumped} = \left(R_{cond,c} + R_{conv}\right) C$$

where

$$C = V \, \rho_s \, c_s$$

```
C=V*rho_s*c_s                    "capacitance of the sensor"
tau=(R_conv+R_cond_c)*C          "time constant of the sensor"
```

The time constant is 0.72 s and the time per cycle (the inerse of the frequency) is 2 s. These quantities are on the same order and therefore it is not likely that the temperature sensor will be able to faithfully follow the reactor temperature.

c.) Develop an analytical model of the temperature response of the sensor.

The temperature sensor is exposed to a sinusoidally varying temperature:

$$T_\infty = \overline{T}_\infty + \Delta T_\infty \sin\left(2 \pi f t\right) \tag{1}$$

The governing equation for the sensor balances heat transfer to ambient against energy storage:

$$0 = \frac{\left[T - T_\infty\right]}{R_{cond,c} + R_{conv}} + C \frac{dT}{dt} \tag{2}$$

Substituting Eq. (1) into Eq. (2) leads to:

$$0 = \frac{\left[T - \overline{T}_\infty - \Delta T_\infty \sin\left(2 \pi f t\right)\right]}{\tau_{lumped}} + \frac{dT}{dt}$$

which is rearranged:

$$\frac{dT}{dt} + \frac{T}{\tau_{lumped}} = \frac{\overline{T}_\infty}{\tau_{lumped}} + \frac{\Delta T_\infty \sin\left(2 \pi f t\right)}{\tau_{lumped}} \tag{3}$$

Equation (3) is a non-homogeneous ordinary differential equation. The solution is assumed to consist of a homogeneous and particular solution:

$$T = T_h + T_p \tag{4}$$

Substituting Eq. (4) into Eq. (3) leads to:

$$\underbrace{\frac{dT_h}{dt}+\frac{T_h}{\tau_{lumped}}}_{\substack{=0\text{ for homogeneous}\\ \text{differential equation}}}+\underbrace{\frac{dT_p}{dt}+\frac{T_p}{\tau_{lumped}}=\frac{\overline{T}_\infty}{\tau_{lumped}}+\frac{\Delta T_\infty \sin\left(2\pi f t\right)}{\tau_{lumped}}}_{\text{particular differential equation}}$$

The homogeneous differential equation is:

$$\frac{dT_h}{dt}+\frac{T_h}{\tau_{lumped}}=0$$

The solution to the homogeneous differential equation can be obtained by separating variables and integrating:

$$\int\frac{dT_h}{T_h}=-\int\frac{dt}{\tau_{lumped}}$$

Carrying out the indefinite integral leads to:

$$\ln\left(T_h\right)=-\frac{t}{\tau_{lumped}}+C_1 \tag{5}$$

where C_1 is a constant of integration. Equation (5) can be rearranged:

$$T_h=\underbrace{\exp\left(C_1\right)}_{C_1^*}\exp\left(-\frac{t}{\tau_{lumped}}\right)=C_1^*\exp\left(-\frac{t}{\tau_{lumped}}\right)$$

where C_1^* is an undetermined constant that will be referred to as C_1:

$$T_h=C_1\exp\left(-\frac{t}{\tau_{lumped}}\right) \tag{6}$$

Notice that the homogeneous solution provided by Eq. (6) dies off after about three time constants.

The particular solution (T_p) is obtained by identifying any function that satisfies the particular differential equation:

$$\frac{dT_p}{dt}+\frac{T_p}{\tau_{lumped}}=\frac{\overline{T}_\infty}{\tau_{lumped}}+\frac{\Delta T_\infty \sin\left(2\pi f t\right)}{\tau_{lumped}} \tag{7}$$

By inspection or using Maple, the sum of a constant and a sine and cosine with the same frequency can be made to solve Eq. (7):

$$T_p = C_2 \sin\left(2\pi f t\right) + C_3 \cos\left(2\pi f t\right) + C_4 \tag{8}$$

Substituting Eq. (8) into Eq. (7) leads to:

$$\begin{aligned} &C_2\, 2\pi f \cos\left(2\pi f t\right) - C_3\, 2\pi f \sin\left(2\pi f t\right) + \frac{C_2}{\tau_{lumped}} \sin\left(2\pi f t\right) \\ &+ \frac{C_3}{\tau_{lumped}} \cos\left(2\pi f t\right) + \frac{C_4}{\tau_{lumped}} = \frac{\overline{T}_\infty}{\tau_{lumped}} + \frac{\Delta T_\infty}{\tau_{lumped}} \sin\left(2\pi f t\right) \end{aligned} \tag{9}$$

Equation (9) can only be true if the constant, sine and cosine terms each separately add to zero:

$$\frac{C_4}{\tau_{lumped}} = \frac{\overline{T}_\infty}{\tau_{lumped}}$$

$$C_2\, 2\pi f + \frac{C_3}{\tau_{lumped}} = 0$$

$$-C_3\, 2\pi f + \frac{C_2}{\tau_{lumped}} = \frac{\Delta T_\infty}{\tau_{lumped}}$$

Solving for C_2, C_3, and C_4 leads to:

$$C_2 = \frac{\Delta T_\infty}{1 + \left(2\pi f \tau_{lumped}\right)^2}$$

$$C_3 = -\frac{2\pi f \tau\, \Delta T_\infty}{1 + \left(2\pi f \tau_{lumped}\right)^2}$$

$$C_4 = \overline{T}_\infty$$

so that the particular solution is:

$$T_p = \overline{T}_\infty + \frac{\Delta T_\infty}{1 + \left(2\pi f \tau_{lumped}\right)^2}\left[\sin\left(2\pi f t\right) - \left(2\pi f \tau_{lumped}\right)\cos\left(2\pi f t\right)\right] \tag{10}$$

and the solution is the sum of the particular and homogeneous solutions, Eqs. (6) and (10):

$$T = C_1 \exp\left(-\frac{t}{\tau_{lumped}}\right) + \overline{T}_\infty + \frac{\Delta T_\infty}{1+\left(2\pi f \tau_{lumped}\right)^2}\left[\sin\left(2\pi f t\right) - \left(2\pi f \tau_{lumped}\right)\cos\left(2\pi f t\right)\right] \quad (11)$$

Note that the same conclusion can be reached using two lines of Maple code; the governing differential equation, Eq. (3), is entered and solved:

```
> restart;
> ODE:=diff(T(t),t)+T(t)/tau=T_infinity_bar/tau+DT_infinity*sin(2*pi*f*t)/tau;
```

$$ODE := \left(\frac{d}{dt}\mathrm{T}(t)\right) + \frac{\mathrm{T}(t)}{\tau} = \frac{T\_infinity\_bar}{\tau} + \frac{DT\_infinity\ \sin(2\,\pi\,f\,t)}{\tau}$$

```
> Ts:=dsolve(ODE);
```

$$Ts := \mathrm{T}(t) = \mathbf{e}^{\left(-\frac{t}{\tau}\right)}\ \_C1 + (T\_infinity\_bar + 4\ T\_infinity\_bar\ f^2\ \pi^2\ \tau^2 - 2\ DT\_infinity\ \cos(2\,\pi\,f\,t)\,f\,\pi\,\tau + DT\_infinity\ \sin(2\,\pi\,f\,t))\ /\ (1 + 4\,f^2\,\pi^2\,\tau^2)$$

The solution identified by Maple is the equivalent to Eq. (11); the solution is copied into EES, with minor editing:

```
"Solution"
Temp = exp(-1/tau*t)*C1-(-T_infinity_bar-4*T_infinity_bar*pi^2*f^2*tau^2+&
            2*DT_infinity*cos(2*pi*f*t)*pi*f*tau&
            -DT_infinity*sin(2*pi*f*t))/(1+4*pi^2*f^2*tau^2)
```

The constant C_1 must be selected so that the boundary condition is satisfied:

$$T_{t=0} = T_{ini} \quad (12)$$

Substituting Eq. (11) into Eq. (12) leads to:

$$T_{t=0} = C_1 + \overline{T}_\infty - \frac{\Delta T_\infty 2\pi f \tau_{lumped}}{1+\left(2\pi f \tau_{lumped}\right)^2} = T_{ini}$$

which leads to:

$$C_1 = \frac{\Delta T_\infty 2\pi f \tau_{lumped}}{1+\left(2\pi f \tau_{lumped}\right)^2} + T_{ini} - \overline{T}_\infty$$

The symbolic expression for the boundary condition can also be found using Maple:

```
> rhs(eval(Ts,t=0))=T_ini;
```

$$\_C1 + \frac{T\_infinity\_bar + 4\,T\_infinity\_bar\,f^2\,\pi^2\,\tau^2 - 2\,DT\_infinity\,f\,\pi\,\tau}{1 + 4 f^2\,\pi^2\,\tau^2} = T\_ini$$

which is copied and pasted into EES:

```
C_1+(T_infinity_bar+4*T_infinity_bar*f^2*pi^2*tau^2-2*DT_infinity*f*pi*tau)/(1+4*f^2*pi^2*tau^2) = T_ini
        "initial condition"
```

The sensor temperature and fluid temperature are converted to Celsius.

```
T_infinity=T_infinity_bar+DT_infinity*sin(2*pi*f*t)          "ambient temperature"
T_C=converttemp(K,C,Temp)                                    "sensor temperature in C"
T_infinity_C=converttemp(K,C,T_infinity)                     "ambient temperature in C"
```

The temperatures are computed in a parametric table in which time set so that it ranges from 0 to 10 s. The fluid and sensor temperature variation are shown as a function of time in Figure 1.



Figure 1: Temperature sensor and fluid temperature as a function of time.

Note that after approximately 3 seconds (i.e., a few time constants) the homogeneous solution has decayed and the temperature response of the sensor is given entirely by the particular solution, Eq. (10):

$$T_p = \overline{T}_\infty + \frac{\Delta T_\infty}{1+\left(2\pi f \tau_{lumped}\right)^2}\left[\sin\left(2\pi f t\right) - \left(2\pi f \tau_{lumped}\right)\cos\left(2\pi f t\right)\right] \tag{10}$$

Equation (10) can be rewritten in terms of an attenuation of the amplitude of the fluid temperature oscillation amplitude (Att) and a phase lag relative to the fluid temperature variation (ϕ)

$$T_p = \overline{T}_\infty + Att\,\Delta T_\infty \sin\left(2\pi f t - \phi\right) \tag{13}$$

Equation (13) is rewritten using the trigonometric identity:

$$T_p = \overline{T}_\infty + Att\,\Delta T_\infty \left[\sin\left(2\pi f t\right)\cos\left(\phi\right) - \cos\left(2\pi f t\right)\sin\left(\phi\right)\right] \qquad (14)$$

Comparing Eq. (13) with Eq. (14) leads to:

$$Att\cos\left(\phi\right) = \frac{1}{1+\left(2\pi f\tau\right)^2} \qquad (15)$$

and

$$Att\sin\left(\phi\right) = \frac{2\pi f\tau_{lumped}}{1+\left(2\pi f\tau_{lumped}\right)^2} \qquad (16)$$

Dividing Eq. (16) by Eq. (15) leads to:

$$\tan\left(\phi\right) = 2\pi f\tau_{lumped}$$

so the phase (the lag) between the sensor and fluid temperature is:

$$\phi = \tan^{-1}\left(2\pi f\tau_{lumped}\right)$$

and the attenuation is:

$$Att = \frac{1}{\sqrt{1+\left(2\pi f\tau_{lumped}\right)^2}}$$

Figure 2 shows the attenuation and phase angle as a function of the product of the frequency and the time constant ($f\tau_{lumped}$).



Figure 2: Attenuation and phase angle as a function of the product of the frequency and the time constant.

Notice that if either the frequency or the time constant is small, then the attenuation goes to unity and the phase goes to zero. In this limit the temperature sensor will faithfully follow the fluid temperature with no error related to the transient response characteristics of the sensor; this is the situation that was illustrated earlier in Figure 3-2(b). In the other limit, if either the frequency or time constant of the sensor are very large then the attenuation will approach zero and the phase will approach $\pi/2$ rad (90 deg.). This situation corresponds to Figure 3-2(c) where the sensor cannot respond to the temperature oscillations. The dynamic characteristics of a temperature sensor are important in many applications and the sensor thermal mass should be carefully considered when selecting an instrument for a transient temperature measurement.